EE 230 Experiment 1

Dhruv Shah

August 1, 2021

1 Introduction

Simulation of various RC circuits is done using NGSPICE in this experiment. The results are then compared with the theoretical values.

2 Procedure

The various circuits and their simulation results are described below,

1. **RC Integrator**



Figure 1: RC Integrator Circuit

The SPICE code for simulation of a RC Integrator circuit with $T = 0.1\tau$ is given below.

```
Dhruv Shah 190020039 RC Integrator Circuit
*Circuit Description
r1 1 2 10k
c1 2 0 0.1u
*Voltage Description
*v1n 1 0 pulse(0 5 delay rise_time fall_time T Period)
v1n 1 0 pulse(0 5 0 0 0 10m 20m)

*transient Analysis
*.tran step 3*period

.tran 0.1m 60m
.control
run
```

```
*Plotting
plot v(1) v(2)
set hcopypscolor = 1
hardcopy integrator_10tau.ps v(1) v(2)
.endc
.end
```



(a) $T = 0.1\tau$ (b) $T = 5\tau$

Figure 2: Waveforms for RC Integrator Circuit

It is noted that the Integrator performs better when $T \gg \tau$.

2. **RC Differentiator**



Figure 3: RC Differentiator Circuit

The SPICE code for simulation of a RC Differentiator circuit with $T = 0.1\tau$ is given below.

```
Dhruv Shah 190020039 RC Differentiator Circuit
*Circuit Description
r1 2 0 10k
c1 1 2 0.1u
*Voltage Description
*v1n 1 0 pulse(0 5 delay rise_time fall_time T Period)
```

```
v1n 1 0 pulse(0 5 0 0 0 0.0001 0.0002)

*transient Analysis
*.tran step 3*period

.tran 1u 4m
.control
run

*Plotting
plot v(1) v(2)
set hcopypscolor = 1
hardcopy diff_pt1tau.ps v(1) v(2)
.endc
.end
```



(a) $T = 0.1\tau$



(b) $T = 10\tau$

Figure 4: Waveforms for RC Differentiator Circuit

It is noted that the Differentiator performs better when $T \ll \tau$.

3. **RC Lowpass Filter**



Figure 5: RC Lowpass Filter Circuit

The SPICE code for simulation of a RC Lowpass filter is given below.

```
Dhruv Shah 190020039 RC Lowpass Filter
*Circuit Description
r1 1 2 10k
c1 2 0 0.1u
*Voltage Description
vin 1 0 dc 0 ac 1

*AC analysis for 1 Hz to 1MHz, 10 points per decade
.ac dec 10 1 1Meg
.control
run

*Magnitude dB plot for v(2) on log scale
plot vdb(2) xlog
set hcopypscolor = 1
hardcopy low_pass.ps vdb(2)
.endc
.end
```



Figure 6: Waveform for RC Lowpass Filter

4. **RC Highpass Filter**



Figure 7: RC Highpass Filter Circuit

The SPICE code for simulation of a RC Highpass filter is given below.

```
Dhruv Shah 190020039 RC Highpass Filter
*Circuit Description
r1 2 0 10k
c1 1 2 0.1u
*Voltage Description
vin 1 0 dc 0 ac 1

*AC analysis for 1 Hz to 1MHz, 10 points per decade
.ac dec 10 1 1Meg
.control
run

*Magnitude dB plot for v(2) on log scale
plot vdb(2) xlog
set hcopypscolor = 1
hardcopy high_pass.ps vdb(2)
.endc
.end
```



Figure 8: Waveform for RC Highpass Filter

5. **RC Bandpass Filter**



Figure 9: RC Bandpass Filter Circuit

The transfer function can be calculated using the principle of voltage divider,

$$\frac{V_{\text{out}}}{V_{\text{in}}} = \frac{sRC}{1 + 3sRC + s^2(RC)^2}$$

At f_0, the gain is maximum. Taking $sRC = \alpha$ gives,

$$\begin{aligned}\frac{V_{\text{out}}}{V_{\text{in}}} &= \frac{\alpha}{1 + 3\alpha + \alpha^2} \\ &= \frac{1}{3 + \alpha + \frac{1}{\alpha}}\end{aligned}$$

The function is maximised at $\alpha = 1$ with a maximum value of $\frac{1}{5}$. This shows that $f_0 = \frac{1}{2\pi RC}$. To calculate -3dB points, we set the gain to $\frac{1}{\sqrt{2}}$ times the maximum gain.

$$\begin{aligned}\frac{\alpha}{1 + 3\alpha + \alpha^2} &= \frac{1}{5\sqrt{2}} \\ \alpha^2 + \left(3 - 5\sqrt{2}\right)\alpha + 1 &= 0 \\ \alpha &= \frac{4.07 \pm \sqrt{16.57 - 4}}{2} \\ \alpha &= 2.03 \pm 1.77\end{aligned}$$

From this we get that,

$$f_0 = \frac{1}{2\pi RC}$$
$$f_l = \frac{1}{2\pi RC}(0.26)$$
$$f_h = f_h = \frac{1}{2\pi RC}(3.8)$$

The SPICE code for simulation of a RC Bandpass filter is given below.

```
Dhruv Shah 190020039 RC bandpass Filter
*Circuit Description
r1 2 0 10k
c1 2 0 0.1u
r2 1 3 10k
c2 3 2 0.1u

*Voltage Description
vin 1 0 dc 0 ac 1

*AC analysis for 1 Hz to 1MHz, 10 points per decade
.ac dec 10 1 1Meg
.control
run

*Magnitude dB plot for v(2) on log scale
plot vdb(2) xlog
set hcopypscolor = 1
hardcopy RC_band_pass.ps vdb(2)

*Measurements
meas ac peak MAX vmag(2)
meas ac fpeak WHEN vmag(2)=peak
let f3db = peak / sqrt(2)
meas ac f1 WHEN vmag(2) = f3db RISE = 1
meas ac f2 WHEN vmag(2) = f3db FALL = 1
.endc
.end
```



Figure 10: Waveform for RC Bandpass Filter

The comparison of simulation results with theory is summarised in the table below,

Sr no.	Quantity	Theoretical	Simulated
1	f_0	159.15	158.49
2	f_l	41.38	48.38
3	f_h	604.79	527.66

RLC Bandpass fiter



Figure 11: RLC Bandpass Filter Circuit

A similar analysis as before shows that,

$$f_0 = \frac{1}{2\pi\sqrt{LC}}$$

$$f_l = \frac{1}{2\pi}\left(-\frac{R}{2L} + \sqrt{\frac{R}{2L}^2 + \frac{1}{LC}}\right)$$

$$f_h = \frac{1}{2\pi}\left(\frac{R}{2L} + \sqrt{\frac{R}{2L}^2 + \frac{1}{LC}}\right)$$

The SPICE code for simulation of a RC Bandpass filter is given below.

```
Dhruv Shah 190020039 RLC Bandpass Filter
*Circuit Description
r1 2 0 1k
l2 1 3 10m
c2 3 2 0.1u

*Voltage Description
vin 1 0 dc 0 ac 1

*AC analysis for 1 Hz to 1MHz, 10 points per decade
.ac dec 10 1 1Meg
.control
run

*Magnitude dB plot for v(2) on log scale
plot vdb(2) xlog
set hcopypscolor = 1
hardcopy RLC_band_pass.ps vdb(2)
*Measurements
meas ac peak MAX vmag(2)
meas ac fpeak WHEN vmag(2)=peak
let f3db = peak / sqrt(2)
meas ac f1 WHEN vmag(2) = f3db RISE = 1
meas ac f2 WHEN vmag(2) = f3db FALL = 1

.endc
.end
```



Figure 12: Waveform for RLC Bandpass Filter

The comparison of simulation results with theory is summarised in the table below,

Sr no.	Quantity	Theoretical	Simulated
1	f_0	5032.92	5011.87
2	f_l	1457.98	1468.64
3	f_h	17373.48	17459.91

3 Major Learnings from this experiment

1. An RC integrator circuit has good performance when $T \gg \tau$.
2. An RC differentiator circuit has good performance when $T \ll \tau$.

4 Challenges Faced

The major challenge faced in this experiment was gaining familiarity with SPICE simulation.

5 Questions and Clarifications

None